



commodore

**Druckerhandbuch
cbm 3022 / 3023**

Inhaltsverzeichnis

1. Allgemeines
2. Inbetriebnahme
3. Wie wird der Drucker angesprochen?
 - 3.1 Was ist ein File?
 - 3.2 Der OPEN-Befehl - Primär- und Sekundäradresse
 - 3.3 Ein/Ausgabebefehle
 - 3.4 Anwendung beim Drucker
 - 3.5 Auslisten von Programmen
4. Spezielle Eigenschaften des Druckers
 - 4.1 Liste der Sekundäradressen
 - 4.2 Sekundäradresse 0 (normaler Ausdruck)
 - 4.3 Sekundäradresse 1 (Daten formatiert ausdrucken)
 - 4.4 Sekundäradresse 2 (Formatierungsanweisungen)
 - 4.5 Sekundäradresse 2 - Liste der Formatierungsanweisungen
 - 4.6 Sekundäradresse 2 - Formatierungsbeispiele
 - 4.7 Sekundäradresse 4 (Fehlermeldungen)
 - 4.8 Sekundäradresse 4 (Liste der Fehlermeldungen)
 - 4.9 Sekundäradresse 5 (Ansteuerung einzelner Punkte in der Druckmatrix)
 - 4.10 Sekundäradresse 6 - Abstand zwischen zwei Zeilen (nur 3022)
5. Steuerbefehle für den Drucker
 - 5.1 Tabelle der Steuerbefehle
 - 5.2 Breitschrift
 - 5.3 Groß/Kleinschrift
 - 5.4 Zeilenvorschub

- 5.5 Zeilen überschreiben
- 5.6 Negativdarstellung
- 5.7 Besonderheiten beim Zeilenvorschub
- 6. Tabelle der ASCII-Zeichen
- 7. Technische Daten
 - 7.1 Einstellen einer anderen Peripheriegeräteadresse
- 8. Beispielprogramme
- 9. Inhaltsverzeichnis

1. Der CBM 3022/3023 - ein leistungsfähiger Drucker

Der CBM 3022 bzw. CBM 3023 stellt einen leistungsfähigen Drucker dar, der in seinem Ausdruck weit über das hinausgeht, was normalerweise auf dem Bildschirm des CBM-Computers möglich ist.

Mit dem Drucker CBM 3022/3023 können Sie alle grossen und kleinen Buchstaben Ihres Rechners ausdrucken. Ebenso sind alle Sonderzeichen möglich. Daneben gibt der Drucker Ihnen auch die Möglichkeit der Negativ-Darstellung.

Den Drucker gibt es in zwei Versionen:

1. als CBM 3023 mit Friktionsantrieb
2. als CBM 3022 mit Traktorführung

Die Traktorführung ist besonders geeignet für vorgedruckte Formulare, Rechnungen usw., also immer da, wo es wichtig ist, daß eine bestimmte Druckposition exakt erreicht wird und wo viele solcher Formulare nacheinander gedruckt werden, ohne daß eine Nachjustierung des Papierses zu erfolgen braucht.

Beide Drucker ermöglichen eine Vielzahl von Formatierungsanweisungen: Vor- und/oder Nachnullen können unterdrückt oder gedruckt werden. Links- und rechtsbündige Darstellung von Zahlen und Texten ist möglich, beim CBM 3022 kann sogar die Anzahl der Zeilen je Seite Ausdruck individuell festgelegt werden. Ein Original und drei Durchschläge werden gleichzeitig gedruckt. Verwendet werden Normalpapiere mit einer Maximalbreite von 21,6cm (8,5 Zoll) zuzüglich jeweils 1,3cm Lochrand. Beide Druckerversionen sind Einwegdrucker und drucken etwa 70 Zeilen zu je 80 Zeichen pro Minute, was einer maximalen Druckrate von 150 Zeichen pro Sekunde entspricht. Die Druckmatrix ist eine 7x6 Matrix, in der sich jeder einzelne Druckpunkt separat ansteuern läßt, was z.B. ausgenutzt werden kann, um eigene Sonderzeichen zu entwerfen bzw. mit hoher Auflösung Kurven in einer Richtung zu plotten.

2. Inbetriebnahme des Druckers

Wenn Sie den Drucker an den Rechner anschließen wollen, dann schalten Sie zuvor bitte unbedingt beide Geräte aus, den Drucker und den Rechner.

Um sicherzustellen, daß Sie, wenn Sie die Steckverbindung herstellen, nicht statisch aufgeladen sind und die Geräte durch statische Entladung beschädigen, greifen Sie bitte kurz an das Gehäuse des Rechners. Sollten Sie "aufgeladen" gewesen sein, dann sind Sie nun "entladen". Dann stecken Sie den Stecker des Druckeranschlußkabels bitte auf die äußerste rechte Anschlußleiste (von hinten gesehen) Ihres Rechners.

Nun können Sie beide Geräte einschalten, zuerst den Rechner, dann den Drucker. Wenn Sie den Drucker zusammen mit der Floppy Disk betreiben wollen, dann ist zuerst die Floppy an die Anschlußleiste des Rechners anzuschließen. Danach wird das Druckeranschlußkabel auf den Anschlußstecker der Floppy aufgesteckt.

3. Wie wird der Drucker angesprochen?

Um den Drucker über den Rechner steuern zu können, muß man zunächst einige grundsätzliche Dinge wissen. Peripheriegeräte, die am IEEE-Bus angeschlossen sind (wie z.B. der Drucker), müssen durch sogenannte Files angesprochen werden.

3.1 Was ist ein File?

Sehr vereinfacht kann man sagen, daß ein File Ihnen die Möglichkeit gibt, ein am PET angeschlossenes Peripheriegerät zu betreiben. Dazu zählen Drucker, Floppy Disk, 2. Cassettenrekorder, aber auch der Bildschirm und der eingebaute Cassettenrekorder lassen sich mittels eines Files ansprechen.

Beim Programmieren eines Files können Sie mehrere Basic-Befehle (OPEN, CLOSE, INPUT, GET, PRINT, CMD, WAIT) verwenden, wichtig sind für den Drucker nur OPEN, CLOSE, PRINT und CMD. Mit dem OPEN-Befehl eröffnen Sie einen File, d.h., Sie steuern das entsprechende Peripheriegerät an.

Hinter dem OPEN-Befehl stehen drei Zahlen und manchmal auch ein in Anführungszeichen (" ") eingeschlossener Text, z.B.

10 OPEN 19,1,2 "Daten 4.1.79". (Ausgabe von Daten über den 1. Cassettenrekorder)

3.2 Der OPEN-Befehl - Primär- und Sekundäradresse

Hinter dem OPEN-Befehl dürfen drei Zahlen folgen:

OPEN A; B, C

Die Buchstaben A, B und C stehen hier für drei Zahlen. Diese Zahlen haben folgende Bedeutung:

1. Zahl (A) : Diese Zahl darf eine beliebige ganze Zahl zwischen 0 und 255 sein.

Sie stellt die File-Nummer dar.

Der Rechner erlaubt max. 10 Files gleichzeitig offen zu halten.

2. Zahl (B) : Diese Zahl bedeutet das Peripheriegerät, welches angesprochen werden soll.

0 = eingebaute Tastatur

1 = 1. Cassettenrekorder

2 = 2. Cassettenrekorder

3 = Bildschirm

4 = Drucker

8 = Floppy Disk

Diese Zahl stellt also die Geräte-Adresse dar und darf im Bereich von 0 bis 15 liegen. Sie wird häufig auch als Primäradresse bezeichnet.

3. Zahl (C): Diese Zahl darf im Bereich von 0 bis 31 liegen. Sie stellt die Sekundäradresse dar. Die Bedeutung dieser Zahl ist von der Primäradresse abhängig, also von dem entsprechenden Peripheriegerät. Für den 1. und 2. Cassettenrekorder hat sie z.B. folgende Bedeutung: 0 = Lesen, 1 = Schreiben mit File-Ende-Zeichen, 2 = Schreiben mit EOT- (end of tape) Zeichen. Für die Sekundäradressen des Druckers siehe Kapitel 4.

Die 2. und die 3. Zahl dürfen in einem OPEN-Befehl fehlen. Der Rechner nimmt in diesem Fall für die 2. Zahl 1 (eins) an, für die 3. Zahl 0 (null).

Beispiele:

OPEN 95,1,Ø 95 : Eröffnen eines Files mit der Nummer 95
 1 : 1. Cassettenrecorder
 Ø : Lesen von Cassette

OPEN 14,4 14 : Eröffnen eines Files mit der Nummer 14
 4 : Drucker
 : Ersatzparameter = Ø (normaler Ausdruck, siehe Kapitel 4)

Hinweis:

Der Rechner erlaubt Ihnen, bis zu 10 Files gleichzeitig offen zu halten. Wenn mehrere Files gleichzeitig offen sind, dann achten Sie bitte darauf, daß Sie verschiedene File-Nummern (1. Zahl) verwenden, sonst erscheint die Fehlermeldung FILE OPEN ERROR!

Bezüglich Primär- und Sekundäradresse gilt diese Einschränkung nicht, es dürfen also mehrere Files gleichzeitig dasselbe Peripheriegerät ansprechen. Bei manchen Peripheriegeräten ist die Anzahl der erlaubten offenen Files, die dieses Gerät dann ansprechen, auf weniger als 10 limitiert, z.B. bei der Floppy Disk. Für den Drucker gilt diese Einschränkung jedoch nicht.

3.3 Ein/Ausgabebefehle

Ist ein File einmal eröffnet, dann können über ihn Werte ausgegeben werden (z.B. über Drucker, Cassettenrekorder) bzw. Werte eingelesen werden (z.B. über Cassettenrekorder). Ist es ein Lese-File, dann können nur Daten gelesen werden (GET, INPUT); ein Ausgaberversuch mittels eines Lese-Files führt zu einer Fehlermeldung. Dies gilt analog für den Ausgabefile.

Die Basic-Befehle GET, INPUT und PRINT sind, wenn sie innerhalb eines Files verwendet werden, mit dem Doppelkreuz (#) zu versehen: PRINT#, GET#, PRINT#.

Hinter das Doppelkreuz wird dann die jeweils zugehörige Filenummer gesetzt, z.B. PRINT#14, GET#95 oder INPUT#95. Hinter die Filenummer kommt dann ein Komma und eine Variable. Beim PRINT-Befehl darf statt der Variablen auch ein in Anführungszeichen (") eingeschlossener Text stehen.

Beispiel:

```
OPEN 95,1,Ø      Lesefile eröffnen
INPUT#95,A      Variable A einlesen
OPEN 14,4,Ø     Schreibfile eröffnen
PRINT#14,X      Variable X ausdrucken
PRINT#14, "ZU"  das Wort ZU ausdrucken
```

Eröffnete Files sind, wenn die entsprechende Programmroutine abgelaufen ist und sie nicht mehr gebraucht werden, stets wieder zu schließen. Dies ist insbesondere dann wichtig, wenn mit sehr vielen Files gearbeitet wird, denn es dürfen ja nicht mehr als 10 gleichzeitig offen sein!

Dazu dient der Basic-Befehl CLOSE.

Hinter CLOSE kommt noch die Nummer des Files, der geschlossen werden soll :

```
CLOSE 95        schließt File 95
CLOSE 14        schließt File 14
```

Der CLOSE-Befehl ist besonders am Ende von Drucker-Files zu setzen. Normalerweise hat ein Drucker einen Pufferspeicher, in dem er die auszudruckenden Werte speichert. Erst wenn der Puffer voll ist, wird gedruckt. Wenn nun der letzte PRINT - Befehl durchgelaufen ist und der Pufferspeicher noch nicht voll ist, dann werden die letzten noch im Pufferspeicher befindlichen Werte nicht ausgedrückt, wenn nicht der Druck-File durch einen CLOSE-Befehl geschlossen wird.

3.4 Anwendung beim Drucker

Sie kennen nun die grundsätzlich notwendigen Befehle, um Ihren Drucker überhaupt ansprechen zu können: Sie eröffnen einen File danach können Sie die Print-Befehle programmieren und wenn diese durchlaufen sind, programmieren Sie noch den CLOSE-Befehl.

Dazu nun noch ein Beispiel: Sie wollen die Zahlen von eins bis zehn untereinander ausdrucken lassen. Das Programm dazu kann dann so aussehen:

```
1Ø OPEN9,4
2Ø FOR I=1 TO 1Ø
3Ø PRINT#9,I
4Ø NEXT
5Ø CLOSE9
```

Zeile 10 eröffnet den Druck-File mit der Nummer 9. Die Ziffer 4 ist die Geräteadresse des Druckers. Die dritte Ziffer darf fehlen, in diesem Fall nimmt der Rechner dafür den Wert 0 an.

Zeile 20 ist der Beginn einer Schleife, die bis Zeile 40 geht und 10 mal durchlaufen wird.

Zeile 30 beinhaltet den Druck-Befehl. Die Schleifenvariable I, die die Werte von 1 bis 10 durchläuft, wird hier jeweils ausgedruckt.

Beachten Sie, daß beim Programmieren der Befehl PRINT#9 nicht durch ?#9 programmiert wird; der Rechner versteht diesen Befehl nicht, es wird nichts gedruckt!

Zeile 40 stellt das Ende der FOR...NEXT-Schleife dar. Die Variable I braucht nicht hinter dem NEXT-Befehl zu stehen (NEXT I ist nicht nötig).

Zeile 50 schließt den Druck-File.

Wie Sie sehen, werden die Druck-Befehle für den Drucker ähnlich programmiert wie es für den Bildschirm notwendig ist, nur daß vorher ein File eröffnet werden muß und die PRINT-Befehle mit dem Doppelkreuz (#) zu versehen sind. Den PRINT-# Befehl können Sie - wie fast alle Basic-Befehle - durch folgende Kurzform programmieren:

30 PRINT # 9, I	darf so programmiert werden:
30 Pr 9, I	was dann beim Auslisten (LIST 30) wieder
30 PRINT # 9,	ergibt!

Folgendes Programm druckt 20 mal GUTEN TAG, ICH BIN DER DRUCKER 3023!

```
10 OPEN 7,4
20 FOR I =1 TO 20
30 PRINT # 7, "GUTEN TAG, ICH BIN DER DRUCKER 3023!."
40 NEXT
50 CLOSE 7
```

3,5 Auslisten von Programmen

Zum Auslisten eines Programmes sind nur wenige Befehle einzugeben:

Als erstes eröffnen Sie einen beliebigen File. Danach tippen Sie den Basic-Befehl CMD (engl. : command = Kommando) ein, sowie die entsprechende File-Nummer. Dann ist der LIST-Befehl einzutippen (LIST oder LIST-40 oder LIST 100

oder LIST 1000-2000) und die Return-Taste zu drücken. Am Drucker leuchtet nun während des Auslistens die Leuchtdiode auf.

Normalerweise wird die eben beschriebene Befehlsfolge über Tastatur eingegeben, aber sie kann auch in das Programm geschrieben werden.

Wenn das Programm ausgelistet ist, dann ist der IEE-Bus noch immer aktiv, denn der CMD-Befehl hat jegliche Ausgabe auf den IEE-Bus geleitet. Sie sehen dies daran, daß die Leuchtdiode noch immer brennt. Um jetzt den Ausgabefile zu schließen, tippen Sie einfach PRINT 1 ein: Die Leuchtdiode erlischt, der Rechner kann wieder wie gewohnt bedient werden.

Die Befehlsfolge zum Auslisten noch einmal kurz zusammengefaßt:

```
OPEN 1,4 : CMD 1 : LIST          eintippen und dann die RETURN-
                                   Taste drücken.
                                   Wenn das Listen beendet ist,
                                   bitte PRINT 1 eintippen, und
                                   die RETURN-Taste drücken!
                                   (LIST listet das gesamte Programm
                                   aus)
```

Noch ein weiteres Beispiel: Die Programmzeilen 1000 bis 2000 sollen ausgelistet werden:

```
OPEN 5,4 : CMD 5 : LIST 1000-2000
```

Zu beachten ist noch, daß das Auslisten von Programmen immer im Groß/Graphik-Modus erfolgt, d.h., daß kleine Buchstaben nicht als Buchstaben, sondern als Sonderzeichen ausgelistet werden. Zu beachten ist auch, daß der Bildschirm des Rechners dunkel wird, wenn der Drucker druckt, als auch beim Listen eines Programmes. Dies ist normal. Die Leuchtdiode, welche beim Arbeiten des Druckers leuchtet, hat noch eine Funktion: Wenn Sie auf die Taste mit der Leuchtdiode drücken, erfolgt ein Papiervorschub (wichtig beim Einlegen des Papiers).

4.1 Liste der Sekundäradressen

Wie im Kapitel 3.2 erklärt wurde, stellt in dem Basic-Befehl OPEN A;B,C die Zahl B die Primäradresse dar, die Zahl C die Sekundäradresse. Die Zahl A stellt die Filenummer dar. Die Primäradresse des Druckers ist normalerweise auf 4 festgelegt, kann aber durch Herstellen einer anderen Lötbrücke auch auf 5, 6 oder 7 festgelegt werden. Hier nun eine kurze Zusammenfassung der Sekundäradressen des CB M 3022/3023 :

- 0 : normaler Ausdruck
- 1 : Daten formatiert ausdrücken
- 2 : Formatierungsanweisungen

- 3 : Zeilenzahl je Seite
- 4 : Fehlermeldungen ermöglichen
- 5 : Ansteuerung einzelner Punkte in der 7 mal 6 Matrix
- 6 : Zeilenabstand (nur bei Traktorführung, nur bei CMB 3022)

Die einzelnen Sekundäradressen werden in folgenden Kapiteln ausführlich erklärt. Grundsätzlich gibt es zwei Arten, wie der Ausdruck des Druckers gesteuert werden kann: Zum einen über die Sekundäradresse, zum anderen über verschiedene Formatierungsanweisungen sowie Anweisungen für Breitschrift, Sonderzeichen, Groß/Kleinbuchstaben, Zeilenzahl je Seite.....

4.2 Sekundäradresse 0 (normaler Ausdruck)

```
OPEN 12,4   oder OPEN 12,4,0
PRINT #12, "Dies ist ein Test"
A = 4 : PRINT #12,A
CLOSE 12
```

erzeugt folgenden Ausdruck:

```
DIES IST EIN TEST
4
```

Beachten Sie, daß die Sekundäradresse im OPEN-Befehl fehlen darf. In diesem Falle wird sie dann als 0 (Null) angenommen. Die Daten (Zahlen oder Text) werden ohne jegliche Formatierung ausgedruckt, in der Reihenfolge der Print-Anweisungen. Sollte hierbei einmal die maximale Länge von 80 Zeichen je Zeile überschritten werden, dann resultiert kein Fehler, sondern der Drucker druckt die überschüssigen Zeichen in die nächste Zeile.

4.3 Sekundäradresse 1 (Daten formatiert ausdrücken)

```
OPEN 6,4,1
PRINT #6,1.35;-.01,178, "ABC" CHR$(29) "DEF"
CLOSE 6
```

kann folgenden Ausdruck erzeugen:

+01.35 0.01- 178 AB DEF

oder:

1.3 .01000- 000178.000 ABC DE

Wie der Ausdruck genau aussieht, lässt sich anhand des oben angegebenen Programms nicht sagen, denn die Sekundäradresse 1 bezieht sich auf eine weitere Sekundäradresse, die Sekundäradresse 2: Es muß noch ein weiterer File eröffnet sein, welcher die Sekundäradresse 2 haben muß, in dem dann die Formatierungsanweisungen stehen, wo also Art und Form des Ausdruckes festgelegt werden. Beachten Sie, daß dieser zweite File nicht schon wieder geschlossen ist, wenn der PRINT #-Befehl (in unserem Beispiel PRINT #6) erreicht wird. Sollte dies trotzdem einmal passieren, dann werden die Daten "normal" ausgedruckt, also ohne jegliche Formatierung, d.h., der Drucker nimmt in diesem Fall die Sekundäradresse 0 (normal Ausdrucken) an. Wenn ein String (Text, Zeichenfolge) ausgedruckt werden soll, und es ist keine gültige Formatierungsanweisung vorhanden, dann werden die Daten ebenfalls wieder ohne Formatierung in der Reihenfolge, wie sie anfallen, ausgedruckt. Beim formatierten Ausdrucken eines Strings ist noch zu beachten, daß das Ende des Strings durch einen Skip-Befehl (engl. skip = springen) gekennzeichnet werden muß. Dieser Sprung-Befehl wird durch CHR # (29) programmiert.

Wenn Sie ein freies alphanumerisches Feld drucken wollen, dann geschieht dies so: CHR # (160) : CHR # (29). CHR # (160) steht für ein "geschiftetes " Blank:

Format AAA AAA AAA

Daten "ABC" CHR # (29) CHR # (160) CHR # (29) "DEF"

Ausdruck ABC DEF

Kurz zusammengefaßt kann man sagen, daß beim formatierten Ausdruck zwei Files eröffnet sein müssen: Erstens der eigentliche Druckfile mit der Sekundäradresse 1 und zweitens der Formatierungsfile mit den entsprechenden Formatierungsanweisungen, der die Sekundäradresse 2 hat.

4.4 Sekundäradresse 2 (Formatierungsanweisungen)

OPEN 7,4,2

PRINT # 7, "AAA 99.99 ###.99 ZZ"

CLOSE 7

• Bitte beachten Sie, daß in diesem Formatierungsfile der

Befehl CLOSE 7 erst stehen darf, wenn der Ausdruck durch den eigentlichen Druckfile erfolgt ist!
Das obrige Beispiel führt in keinem Fall zu dem Ausdruck .

```
AAA 99.99 $$$.99 ZZ
```

wie man vielleicht meinen könnte, sondern legt nur die Art und Weise eines zukünftigen Ausdrucks fest, ausgedruckt werden kann nur über einen zweiten File mit der Sekundäradresse 1 !
Beispiel dazu:

```
10 OPEN 7,4,2
```

```
20 PRINT # 7, "AAA 99.99 $$$.99 ZZ"
```

```
30 OPEN 5,4,1
```

```
40 X = 3.5:Y = 18:Z = 3
```

```
50 PRINT # 5, "ABCDEF" ;X;Y;Z
```

```
60 CLOSE 5:CLOSE 7
```

Der Ausdruck sieht dann so aus:

```
ABC 3.50 18.00 03
```

Warum der Ausdruck so erfolgt, ist an dieser Stelle noch etwas schwer zu verstehen. Schauen Sie sich dazu bitte die nun folgenden Formatierungsbefehle an!

4.5 Sekundäradresse 2 - Liste der Formatierungsbefehle

Die einzelnen Formatierungsbefehle kann man in drei Gruppen unterteilen:

Numerische Anweisungen: 9Z \$. -

Alphanumerische Anweisungen A

Skip (Sprung) CHR \$(29)

9 - steht für eine Ziffer in einem numerischen Feld. Wenn keine Ziffer vorhanden ist, dann wird im Vorkommaanteil ein Blank gedruckt, im Nachkommaanteil werden Nullen gedruckt.

Z - steht ebenfalls für eine Ziffer in einem numerischen Feld. Im Gegensatz zu 9 werden aber hier an den Vorkommastellen, auf die keine Ziffern entfallen, Nullen gedruckt.

\$ - Wenn mit dem \$ -Zeichen formatiert wird, dann wird das Feld als ein Feld mit Dollar-Beträgen betrachtet, und vor jede auszudruckende Zahl wird das \$ -Zeichen gesetzt. Wird nur mit

mehreren # -Zeichen formatiert, dann wird das # -Zeichen unmittelbar vor die auszudruckende Zahl gesetzt. Wird jedoch mit einem # -Zeichen und mehreren 9-Zeichen formatiert, dann steht das # -Zeichen zu Beginn des Feldes und es folgen - je nach Formatierung mit der 9 - kein, ein oder mehrere Blanks, bis die erste Ziffer gedruckt wird.

S - Wenn das S (S für engl. sign = Zeichen) zu Beginn eines Formatierungsfeldes steht, dann wird in jedem Fall das Vorzeichen der Zahl mit ausgedruckt, also entweder ein + oder ein - .Das S darf nur am Anfang einer Formatierungsanweisung stehen, nicht innerhalb des Formatierungsfeldes. Mehrere S dürfen nicht am Anfang einer Formatierungsanweisung stehen.

" - Der . legt die Positionen des Kommas fest und wird auch dann ausgedruckt, wenn er in der Formatierungsanweisung steht und kein Nachkommaanteil festgelegt ist!

- - Das - steht für ein Minuszeichen, welches, im Gegensatz zu S, hinter der auszudruckenden Zahl gedruckt wird. Bei einer positiven Zahl wird allerdings kein Pluszeichen hinterher gedruckt. Wenn sowohl das S als auch das - in einer Formatierungsanweisung auftauchen, dann führt dies zu einem Fehler.

A - steht für ein alphanumerisches Zeichen, einen Buchstaben z.B..Sollten dem String, welcher auszudrucken ist, Blanks vorausgehen, dann werden sie abgeschnitten: Der String wird linksbündig gedruckt! Eine Ausnahme bildet hier jedoch das geshiftete Blank: CHR # (160), welches nicht gelöscht wird, auch wenn es zu Beginn des Strings steht. Maximal 80 Zeichen dürfen die Formatierungsanweisungen für eine Druckzeile umfassen, weil nur 80 Zeichen/Zeile im Ausdruck möglich sind. Formatieren Sie mehr, dann werden die überschüssigen Zeichen in die nächste Zeile gedruckt. Stehen im Druckbefehl mehr Buchstaben als die Formatierung erlaubt, dann werden die überschüssigen Buchstaben abgeschnitten, es resultiert also kein Fehler!

Es ist möglich, daß Sie sich ein individuelles Ausdrucksmuster erstellen wollen. Dazu können Sie z.B. die Sonderzeichen, also die graphischen Zeichen verwenden. Auf Graphik schalten Sie durch CHR # (145) um (nur Drucker), diese Umschaltung können Sie durch CHR # (17) wieder rückgängig machen. Andererseits haben Sie auch die Möglichkeit, ein eigenes Drucksymbol innerhalb der 7 mal 6 Matrix zu erstellen. Dies wird bei der Beschreibung der Sekundäradresse 5 erklärt.

4.5 Sekundäradresse 2 - Formatierungsbeispiele

Formatierungsfeld	Daten	Ausdruck
AAAAA	ABC	ABC
AAAAA	ABCDEFG	ABCDE
#####	85	# 85
#####	12345	*****
# 9999	85	# 85
# 9999	-1234	# 1234
# 9999	12345	*****
# 99.99	76	# 76.00
# 99.99	-72	# 72.00
# 99.99-	-72.5	# 72.50-
# 99.99-	72.5	# 72.50
S # 99.99	1.2	+ # 1.20
ZZZZ	14	0014
ZZ	1979	**
ZZ. 999	14	14.000
ZZZ.999	14	014.000
SZZZ.999	14	+014.000
999.99	14	14.00
.99	14	..**
.99	.017	.01
S.999	1.5E-02	+.015
Z.999-	1.5E-02	0.015
SZ.999-	1.5E-02	+0.015

und BAD FORMAT
Meldung

4.6 Sekundäradresse 3 (Zeilen pro Seite)

Die Seitenanzahl je Ausdruckseite kann variabel eingestellt werden. Diese Funktion des Druckers wird durch CHR # (147) eingeschaltet und durch CHR # (19) abgeschaltet. Wenn die Funktion eingeschaltet ist, dann ist die Zeilenanzahl je Druckseite auf 66 festgelegt: 3 Zeilen oberer Rand plus 60 Zeilen für den Ausdruck plus 3 Zeilen unterer Rand.

Sie sind aber nicht an die Zahl von 60 Zeilen pro Seite gebunden, sondern können diese Zahl variabel festsetzen: wollen Sie z.B. nur 20 Zeilen pro Seite, dann kann dies so programmiert bzw. eingetippt werden:

```
OPEN 9,4,3
PRINT # 9,20
```

Wenn Sie eine Seite abschließen wollen, dann geben Sie CHR # (19) ein, z.B. PRINT # 14, CHR # (19): wenn noch nicht alle Zeilen der Seite beschrieben waren, dann erfolgt für jede noch nicht beschriebene Zeile ein Zeilenvorschub.

4.7 Sekundäradresse 4 (Fehlermeldungen)

Wenn es zu einem Formatierungsfehler gekommen ist, dann bestehen für den Ausdruck zwei Möglichkeiten: entweder druckt der Drucker eine entsprechende Fehlermeldung aus oder nicht. In jedem Fall jedoch wird, sobald ein solcher Formatierungsfehler aufgetreten ist, die Sekundäradresse des Formatierungsfiles auf Null gesetzt, d.h. es wird nicht mehr formatiert ausgedruckt.

Wenn Sie den Ausdruck von Fehlermeldungen wünschen, dann können Sie dies durch Programmieren eines "Fehlerfiles" ermöglichen. Solange kein Formatierungsfehler auftritt, ist dieser "Fehlerfile" für den Ausdruck bedeutungslos.

Der "Fehlerfile" wird so programmiert:

```
1Ø OPEN 9,4,4
2Ø PRINT #9
```

und ermöglicht dann eine der im Anschluß folgenden Fehlermeldungen. Gleichgültig ob Sie den "Fehlerfile" eröffnen oder nicht, kann es in einem numerischen Feld zu einem Fehler kommen. Ein solcher Fehler wird im Ausdruck durch das Multiplikationszeichen (*) angezeigt, d.h. das entsprechende Feld wird mit diesen Zeichen aufgefüllt und ausgedruckt. Dieser Fehler hat aber sonst keine weiteren Folgen, d.h. die Formatierungsaneisungen bleiben nach wie vor gültig. Solange der Fehler nicht mehr auftritt, wird normal weitergedruckt und formatiert.

4.8 Sekundäradresse 4 - Liste der Fehlermeldungen

**** BAD FORMAT ****	falsches Formatierungszeichen
**** DATA-FORMAT MISMATCH ****	numerische Daten statt alphanumerische oder umgekehrt
**** BAD NUMERIC DATA ****	falsche numerische Daten
**** EXCESS LINES PER PAGE ****	zu viele Zeilen je Seite
**** BAD COMMAND ****	falscher Befehl


Beispiel für einen Fehlerausdruck:

```
99 Ø1
ZZ ZZ XZ ZZ
```

```
**** BAD FORMAT ****
(Das Formatierungszeichen X gibt es nicht)
```

4.9 Sekundäradresse 5 (Ansteuerung einzelner Punkte in der 7x6 Druckmatrix)

Sie können ein Drucksymbol frei zusammenstellen, denn der Drucker gibt Ihnen die Möglichkeit, jeden einzelnen Punkt in der 7x6 Matrix anzusteuern. Sie eröffnen dazu ein File mit der Sekundäradresse 5, durch welchen die Ansteuerung der einzelnen Punkte festgelegt wird. Innerhalb einer Ausdruckzeile kann dieses Sondersymbol nicht mehr verändert werden. Den Ausdruck dieses Zeichens veranlassen Sie durch CHR\$(254), z.B. PRINT#4, CHR\$(254).



es empfiehlt sich, eine solche Zeichnung zu erstellen und darin die gewünschten Druckpunkte einzuzichnen. Die Binärwerte der Matrix berechnen Sie so wie in der Zeichnung.

Hierzu ein Beispielprogramm :

```

@ Commodore Business Machines

```

```

10 DATA28,34,65,65,54,34
20 OPEN5,4,5
30 FORI=1TO6:READA:A$=A$+CHR$(A):NEXT
40 PRINT#5,A$
50 OPEN4,4
60 FORI=1TO10
70 PRINT#4,CHR$(1)CHR$(254)" COMMODORE BUSINESS MACHINES"
80 NEXT
85 CMD4:LIST
90 REM CHR$(1) = SPERRSCHRIFT
95 REM CHR$(254) = MATRIXSONDERZEICHEN DRUCKEN
100 REM "↑" KURSOR NACH OBEN = GROSSBUCHSTABEN
110 REM "↓" KURSOR NACH UNTEN = KLEINBUCHSTABEN
READY.

```

Erklärung des "Matrixprogrammes"

Zeile 10 enthält die Daten für die einzelnen Matrixpunkte.
Zeile 20 eröffnet den Matrixfile.
Zeile 30 wandelt die Zahlen der Matrix in Zeichen um und speichert sie in A # .
Zeile 40 gibt die in A # gespeicherte Matrix an den Drucker aus, ohne daß jetzt schon etwas ausgedruckt wird. Der Drucker weiß nur, was er ggf. drucken soll.
Zeile 50 eröffnet den eigentlichen Ausdruckfile.
Zeile 60 enthält eine FOR...NEXT Schleife: es soll 10x gedruckt werden, was in Zeile 70 steht.
Zeile 70 beinhaltet die Ausdruckbefehle: CHR # (1) ist Breitschrift, CHR # (254) veranlasst den Ausdruck der erstellten Matrix, die Sonderzeichen im Text stellen auf Groß/Kleinschreibung um (nur auf dem Drucker, auf dem Rechner geht dies durch POKE 59468,14 bzw. durch POKE 59468,12)
Zeile 80 stellt das Ende der in Zeile 60 eröffneten FOR...NEXT Schleife dar.

4.10 Sekundäradresse 6

Mittels dieser Sekundäradresse kann die Schrittzahl zwischen zwei Zeilen verändert werden (nur CBM 3022).

Siehe Anhang:

5. Steuerbefehle für den Drucker

Durch verschiedene Befehle kann der Drucker beeinflusst werden, ob er große oder kleine Buchstaben, ob er Breitschrift oder Normalschrift schreiben soll usw. Alle diese Befehle werden in den eigentlichen Ausdruck-PRINT # -Befehl einprogrammiert, in den PRINT # -Befehl also, dessen File die Sekundäradresse 0 oder 1 hat. Hinter den PRINT # -Befehl kommt noch die Filenummer und dann CHR # (...). In der Klammer steht eine Zahl, die den eigentlichen Steuerbefehl beinhaltet. Den Ausdruck des Matrixsonderzeichens kann man so durch PRINT # 4, CHR # (254) wie im vorigen Kapitel beschrieben erreichen.

5.1 Tabelle der Steuerbefehle

Bedeutung	Code	ASCII	Computer
Breitschrift	1	SOH	CHR # (1)

Bedeutung	Code	ASCII	Computer
Normalschrift	129		CHR # (129)
Zeilen/Seite EIN u. Rücksetzen	147		CLR
Zeilen/Seite AUS	19	DC3	HOME
Negativdarstellung EIN	18	DC2	RVS
Negativdarstellung AUS	146		OFF
Wagenrücklauf und Zeilenvorschub	13	CR	CR
Wagenrücklauf ohne Zeilenvorschub	141		CHR # (141)
Zeilenvorschub	10	LF	CHR # (10)
Graphik/Großbuch- staben	145		CHR # (145)
Groß/Kleinschreibung	17		CHR # (17)
Skip (Ende Zeichen- kette bei formatier- tem Ausdruck)	29		CHR # (29)
Matrixsonderzeichen	254		CHR # (254)
Blank (geshiftet)	160		CHR # (160)

5.2 Breitschrift

Normalerweise erfolgt der Ausdruck innerhalb einer 7x6 Matrix, wobei dann 80 Zeichen pro Zeile möglich sind. Breitschrift bedeutet, daß diese 7x6 Matrix in eine 7x12 Matrix umgewandelt wird, wobei dann logischerweise nur noch 40 Zeichen pro Zeile möglich sind, weil die Breite der einzelnen Zeichen auf das Doppelte gestiegen ist. Auf Breitschrift schalten Sie um durch CHR # (1), auf Normalschrift durch CHR # (129).
Hier der Buchstabe A in Normalschrift und in Breitschrift:



Hier nun ein kleines Programm für Normalschrift und für Breitschrift:

```
10 OPEN 8,4
20 PRINT #8, CHR $(1);"DIES IST BREITSCHRIFT"
30 PRINT #8, CHR $(129);"DIES IST NORMALSCHRIFT"
40 CLOSE 8
```

5.3 Groß / Kleinschrift

Ihren Computer können Sie auf große und kleine Buchstaben durch POKE 59468,14 umstellen. Beim Drucker geschieht dies durch den Befehl PRINT #8, CHR \$(17), wenn vorher der File 8 durch OPEN 8,4 z.B. eröffnet wurde.

Ob die einprogrammierten Buchstaben nun als große oder als kleine Buchstaben ausgedruckt werden, das wird durch ein jeweils vor den Buchstaben in die PRINT# -Anweisung einzubauendes Sonderzeichen gesteuert:

Groß: Sonderzeichen für Cursor nach oben = CHR \$(145)

Klein: Sonderzeichen für Cursor nach unten = CHR \$(17)

Hier nun ein Beispiel dazu:

```
10 OPEN 8,4
20 PRINT #8, CHR $(145)"NUR GROSSE BUCHSTABEN"
30 PRINT #8, CHR $(17)"NUR KLEINE BUCHSTABEN"
40 CLOSE 8
```

Der Ausdruck sieht dann so aus:

```
NUR GROSSE BUCHSTABEN
nur kleine buchstaben
```

Ein weiteres Beispiel dazu befindet sich in Kapitel 4.7: das Programm, welches das Matrixsonderzeichen ausdrückt.

Wenn Sie Ihren Computer auf Groß/Kleinschreibung umstellen, dann wird jeweils der gesamte Bildschirm umgestellt. Beim Drucker ist dies genauso, doch können Sie dort sofort auch wieder auf Großschreibung/Graphik umschalten und so - im Gegensatz zum Bildschirm des Computers - abwechselnd kleine Buchstaben und graphische Zeichen auf die selbe Seite ausdrucken. Auf Groß/Graphik schalten Sie durch CHR \$(145), auf Groß/Kleinschreibung durch CHR \$(17), jeweils in die entsprechende PRINT-Anweisung integriert, wie oben im Beispiel gezeigt.

Sowohl bei den Rechnern der 3000er Serie als auch beim 2001 sollte der Bildschirm mit POKE 59468,14 auf Groß/Kleinschrift umgestellt werden, damit der Bildschirm das gleiche anzeigt, was der Drucker im entsprechenden Modus (Sonderzeichen Cursor nach unten stellt den Drucker auf Groß/Kleinschrift um, Sonderzeichen Cursor nach oben stellt den Drucker auf Groß/Graphik um) anzeigt.

Eines ist jedoch zu beachten: beim 2001 in Verbindung mit dem cbm 3022/3023 stimmt der Ausdruck mit dem Bildschirm nicht ganz überein; große Buchstaben auf dem Bildschirm werden als kleine Buchstaben vom Drucker gedruckt, bei kleinen Buchstaben ist es umgekehrt.

5.4 Zeilenvorschub

In einem Programm kann es durchaus vorkommen, daß Sie eine oder mehrere Leerzeilen drucken möchten. Dies geht z.B. dadurch, daß Sie einen "leeren" PRINT #-Befehl programmieren, gleiches erreichen Sie auch durch CHR #(13).

```
1Ø OPEN 8,4 : REM "Druckfile eröffnen"
2Ø PRINT #8:PRINT 8 : REM "2 Leerzeilen drucken"
3Ø PRINT #8,CHR #(13);CHR#(13) : REM "2Leerzeilen drucken"
4Ø FOR I=1TO19:PRINT 8:NEXT : REM"19 Leerzeilen drucken"
5Ø CLOSE 8 : REM" Druckfile schließen"
```

5.5 Zeilen überschreiben



In seltenen Fällen kann es erwünscht sein, eine Zeile ganz oder teilweise zu überschreiben. Dies geschieht unter Zuhilfenahme von CHR #(141): Wagenrücklauf ohne Zeilenvorschub. Aber eines ist dabei unbedingt zu beachten: hinter dem letzten PRINT #-Befehl der zu überschreitenden Seite muß immer ein Semikolon stehen, zum Zeichen, daß die Zeile noch nicht abgeschlossen ist und um einen Zeilenvorschub zu unterdrücken. Auch hierzu ein Beispiel:

```
1Ø OPEN 8,4
2Ø PRINT # 8,"123456789Ø123456789Ø";
3Ø PRINT # 8,CHR #(141);"ABC"
4Ø CLOSE 8
```

In diesem Beispiel werden die ersten drei Ziffern (123) durch ABC überschrieben.

5.6 Negativdarstellung

Buchstaben können "negativ" ausgedruckt werden. Die Druckmatrix wird dabei invertiert. Das Bild macht dies deutlich. Zu beachten ist dabei, daß dadurch die Nadeln des Druckkopfes extrem belastet werden, was deren Lebensdauer wesentlich verkürzt.



5.7 Besonderheiten beim Zeilenvorschub

Wenn der Wagenrücklauf/Zeilenvorschub-Befehl ausgeführt wird, dann werden sowohl Negativdarstellung als auch Sperrschrift als auch Hochkommamodus zurückgesetzt und müssen für die nächste Zeile neu programmiert werden.

6. Tabelle der ASCII-Zeichen

Siehe Tabelle im Handbuch Personal Electronic Transactor 2001 bzw. cbm 3001.

7. Technische Daten

Bausteine des Druckers:	6504 Mikroprozessor (2) 6532 Zeitgeber, RAM (2) 6114 1k x 4 RAM 6332 4k x 8 ROM
Mechanik:	Modell 3110 von Epson, Druckkopf DH-70 von Epson
Leistungsaufnahme:	100 Watt bei 120/220/240 Volt, 50 oder 60 Hz
Interface:	IEEE-488 24-polige Standardsteckverbindung Gerätenummer zwischen 4 - 11 wähl- bar (Brücke) nur Listenerfunktionen, erkennt Sekundäradressen
Zeichenvorrat:	ASCII-Groß- und Kleinbuchstaben sowie cbm Graphik-Zeichen Sperrschrift, frei programmierbare Sonderzeichen
Datenformatierung:	Vornullenausdruck möglich Festlegung der Anzahl der Nachkomma- stellen; links- und rechtsbündiger Ausdruck; Textausdruck linksbündig Dollarzeichen
Druckmethode:	Nadeldruck
Matrix:	7x6 bzw. 7x12 (Sperrschrift)
Geschwindigkeit:	etwa 70 Zeilen pro Minute 150 Zeichen pro Sekunde
Druckweg:	Einrichtungsdruck, vorwärts
Zeichen pro Zeile:	max. 80 Zeichen pro Zeile bei 7x6 Mat. max. 40 Zeichen pro Zeile bei 7x12 Mat.
Zeilenabstand:	4.2 mm
Leerraum zwischen zwei Zeilen:	1.5 mm
Zeichengröße:	2.7 mm hoch und 2.2 mm breit
Papierart:	Normalpapier beim cbm 3023 gefaltetes Papier beim cbm 3022 gefaltetes Papier mit Lochrand

Papierabmessung: 254 mm + 1.5 mm
- 2.0 mm (cbm 3023)
216 mm + 1.5 mm
- 2.0 mm plus Lochrand ca.
130 mm (cbm 3022)

Papierdicke: 0.07 bis 0.09 mm

Papiergewicht: 52 bis 64 g/Quadratzenimeter

Farbband: einfarbig
Nylonfaserband mit Stop-Löchern,
Bandendekontrolle
Breite: 13 mm/Länge 11 m
Lebensdauer: 50 Millionen Zeichen bei
25 Grad Celsius und ca. 14 Nadelan-
schlägen je Zeichen
Verwendung anderer Farbbänder kann die
Lebensdauer des Druckkopfes verkürzen!

Farbbandspule: Breite 13 mm, Durchmesser 50 mm

Abmessungen cbm 3023: Breite: 34 cm
Länge: 17,7 cm
Höhe: 9,4 cm
Gewicht: 3 kg

Abmessungen cbm 3022: Breite: 38,8 cm
Länge: 21,5 cm
Höhe: 17,0 cm
Gewicht: 3,5 kg
Motor: 4 Phasen Impulsmotor

Zeilenabstand cbm 3022: programmierbar (4.2 mm bei 24 Impul-
sen und 3.2 mm bei 18
Impulsen)

Papiervorschub cbm 3022: 10 Zeilen/sec.

Lebensdauer des Druck-
kopfes cbm 3022/3023: 60 Millionen Zeichen bei durchschnitt-
lich 14 Nadelanschlägen je Zeichen.

7.1 Einstellen einer anderen Peripheriegeräteadresse

Werkseitig ist Ihr Drucker auf die Peripheriegeräteadresse 4 eingestellt. Unter bestimmten Sonderbedingungen (Anschluß von Meßgeräten z.B.) kann es erforderlich sein, dem Drucker eine andere Peripheriegeräteadresse zu geben. Sie sollten dies nicht selbst tun, denn durch nicht fachgerechte Eingriffe am Drucker verlieren Sie die Garantie. Ihr Fachhändler wird dies jedoch gerne für Sie tun. Hier die dazu notwendige Tabelle:

Ø = offen
1 = Brücke

Peripheriegeräte-Nr.	123
4	111
5	Ø11
6	1Ø1
7	ØØ1
8	11Ø
9	Ø1Ø
10	1ØØ
11	ØØØ

Beispiel 1

ALPHANUMERISCHES FORMATIERUNGSBEISPIEL

```
AA AA AA AA
ABC
AB AB AB AB
```

```
10 OPEN1,4 :REM NORMALER AUSDRUCK
20 PRINT#1,"ALPHANUMERISCHES FORMATIERUNGSBEISPIEL":FORI=1TO10:PRINT#1:NEXT
30 OPEN2,4,1: REM FORMATIERTER AUSDRUCK
40 OPEN3,4,2: REM FORMATIERUNGSFILE
50 OPEN4,4,4: REM ERMOEGLICHT FEHLERAUSDRUCK
60 PRINT#4
70 A$="AA AA AA AA": REM FORMATIERUNGSSTRING
80 B$="ABC" : REM DER FORMATIERT AUSZUDRUCKENDE STRING
90 PRINT#1,A$:REM DRUCKT FORMATIERUNGSANWEISUNG AUS
100 PRINT#1,B$:REM FORMATIERT AUSZUDRUCKENDEN STRING
110 PRINT#3,A$:REM FORMATIERUNG FESTGELEGT
120 REM SKIP = CHR$(29) MUSS EINGEGEBEN WERDEN UM DAS ENDE DES STRINGS
130 REM ZU KENNZEICHNEN; STRING=ZEICHENKETTE, HIER AUS ABC BESTEHEND
140 C$=CHR$(29)
150 PRINT#2,C$B$C$B$C$B$C$B$
160 CMD1
170 PRINT:PRINT:PRINT
180 LIST
READY.
```

Beispiel 2

VORNULLEN

+0001.	+0002.	0003.	0004
+0002.	+0003.	0004.	0005
+0003.	+0004.	0005.	0006
+0004.	+0005.	0006.	0007
+0005.	+0006.	0007.	0008
+0006.	+0007.	0008.	0009
+0007.	+0008.	0009.	0010
+0008.	+0009.	0010.	0011
+0009.	+0010.	0011.	0012
+0010.	+0011.	0012.	0013

```
10 OPEN1,4 :REM NORMALER AUSDRUCK
15 PRINT#1," VORNULLEN":FORI=1TO8:PRINT#1:NEXT
20 OPEN2,4,1 :REM FORMATIERTER AUSDRUCK
30 OPEN3,4,2 :REM FORMATIERUNGSGFILE
40 OPEN4,4,4 :REM ERMOEGLICHT FEHLERAUSDRUCK
50 PRINT#4 :REM ERMOEGLICHT FEHLERAUSDRUCK
60 F$="SZZZZ. SZZZZ. ZZZZ. ZZZZ":REM FORMATIERUNGSANWEISUNG
70 PRINT#3,F$
80 FORI=1TO10
90 PRINT#2,I,I+1,I+2,I+3
100 NEXT
110 CMD1
120 PRINT:PRINT:PRINT:PRINT
130 LIST
READY.
```

Beispiel 3

1 - 1 1-

```
10 OPEN1,4 :REM NORMALER AUSDRUCK
20 OPEN2,4,1 :REM FORMATIERTER AUSDRUCK
30 OPEN3,4,2 :REM FORMATIERUNGSFILE
40 OPEN4,4,4 :REM ERMOEGLICHT FEHLERAUSDRUCK
50 PRINT#4 :REM ERMOEGLICHT FEHLERAUSDRUCK
60 F$="999 S999 999-":REM FORMATIERUNGSANWEISUNG
70 PRINT#3,F$
90 PRINT#2,-1,-1,-1
130 CMD1
140 PRINT:PRINT:PRINT:PRINT
150 LIST
READY.
```

Das "-", welches hinter einer 9 in der Formatierungsanweisung stehen darf, druckt hinter der Zahl ein Blank aus, wenn es eine positive Zahl ist, und ein "-", wenn es eine negative Zahl ist.

Haben Sie mit einem "S" formatiert, wie hier in Zeile 60, dann wird das Vorzeichen in jedem Fall gedruckt, entweder "+" oder "-".

Wenn Sie weder "S" noch "-" in die Formatierungsanweisung geschrieben haben, dann wird die Zahl ohne Vorzeichen ausgedruckt, auch wenn Sie negativ ist.

Beispiel 4

NACHKOMMASTELLEN

5.00	-	1.25000	3
10.00	-	2.50000	6
15.00	-	3.75000	9
20.00	-	5.00000	12
25.00	-	6.25000	15
30.00	-	7.50000	18
35.00	-	8.75000	21
40.00	-	10.00000	24
45.00	-	11.25000	27
50.00	-	12.50000	30

```
10 OPEN1,4 :REM NORMALER AUSDRUCK
15 PRINT#1,"      NACHKOMMASTELLEN":FORI=1TO8:PRINT#1:NEXT
20 OPEN2,4,1 :REM FORMATIERTER AUSDRUCK
30 OPEN3,4,2 :REM FORMATIERUNGSDATEI
40 OPEN4,4,4 :REM ERMOEGLICHT FEHLERAUSDRUCK
50 PRINT#4 :REM ERMOEGLICHT FEHLERAUSDRUCK
60 F$="999.99  S999.99999  999-":REM FORMATIERUNGSANWEISUNG
70 PRINT#3,F$
80 FORI=1TO10
90 PRINT#2,-10*I/2,-5*I/4,3*I
100 NEXT
110 CMD1
120 PRINT:PRINT:PRINT:PRINT
130 LIST
140 REM SIND IN DER FORMATIERUNGSANWEISUNG MEHR NACHKOMMASTELLEN VORGESEHEN,
150 REM ALS DIE AUSZUDRUCKENDE ZAHL BESITZT, DANN WERDEN DIE FREIEN STELLEN
160 REM MIT NULLEN AUFGEFUELLT
170 REM HAT DIE ZAHL MEHR NACHKOMMASTELLEN, DANN WERDEN DIESE ABGESCHNITTEN
READY.
```

Beispiel 5

FESTES UND FLIESENDES \$ - FELD

\$ 40.68	\$8.08	\$9486	\$.4930
\$ 70.40	\$9.45	\$3402	\$.7012
\$ 46.70	\$2.84	\$ 681	\$.0853
\$ 66.84	\$7.82	\$7409	\$.8922
90.38	\$1.84	\$7160	\$.6314
\$ 26.92	\$.67	\$6562	\$.6934
\$ 69.66	\$5.35	\$6795	\$.2957
\$ 26.66	\$6.08	\$3503	\$.9311
\$ 38.93	\$8.04	\$7381	\$.0786
\$ 8.94	\$5.25	\$6516	\$.2912

```

10 OPEN1,4 :REM NORMALER AUSDRUCK
15 PRINT#1,"      FESTES UND FLIESENDES $ - FELD":FORI=1TO8:PRINT#1:NEXT
20 OPEN2,4,1 :REM FORMATIERTER AUSDRUCK
30 OPEN3,4,2 :REM FORMATIERUNGSGFILE
40 OPEN4,4,4 :REM ERMOEGLICHT FEHLERAUSDRUCK
50 PRINT#4 :REM ERMOEGLICHT FEHLERAUSDRUCK
60 F$="$999.99      $$$$$.99  $9999  $.9999
70 PRINT#3,F$
80 FORI=1TO 10
90 PRINT#2,RND(1)*100;RND(1)*10;RND(1)*10000;RND(1)
100 NEXT
130 CMD1
140 PRINT:PRINT:PRINT:PRINT
150 LIST
160 REM IN FESTEN $ - FELDERN WIRD DAS $ - ZEICHEN AUF DIE AUSSERSTE LINKE
170 REM DRUCKPOSITION GESETZT, FREIE STELLEN WERDEN ALS BLANKS GEDRUCKT.
180 REM IN FLIESENDEN $ - FELDERN WIRD ES DIREKT VOR DIE ERSTE ZIFFER BZW.
190 REM DEN DEZIMALPUNKT GESETZT, NICHT BENUTZTE FORMATIERUNGSGSTELLEN
200 REM BLEIBEN FREI.
READY.
```

Beispiel 6

BEISPIEL FUER GEMISCHE FORMATIERUNG


000681	CBM 3022/3023	\$3000.00	5
000681	ANLEITUNG	\$.00	
892013	ARMIN WEINGAR	\$.17	66
000017	STEINACKER	\$.17	66
006601	BISCHMISHEIM	\$66.01	17

```

10 OPEN1,4 :REM NORMALER AUSDRUCK
15 PRINT#1," BEISPIEL FUER GEMISCHE FORMATIERUNG":FORI=1TO8:PRINT#1:NEXT
20 OPEN2,4,1 :REM FORMATIERTER AUSDRUCK
30 OPEN3,4,2 :REM FORMATIERUNGSFILE
40 OPEN4,4,4 :REM FUER FEHLERDIAGNOSE
50 PRINT#4 :REM ERMOEGLICHT GGF. FEHLERAUSDRUCK
60 F$="ZZZZZ AAAAAAAAAAAAAA $$$$$$.99 999"
70 PRINT#3,F$
80 A1=1:A2=681:A3=892013:A4=17:A5=6601
90 B1$="CBM 3022/3023":B2$="ANLEITUNG":V3$="ARMIN":N3$="WEINGARD"
100 S1$="STEIN":S2$="ACKER":O1$="BISCH":O2$="MISHEIM"
110 C1=3000:C2=0:C3=660:C4=.17:C5=66.01
120 D1=5:D2=0:D3=33:D4=66.01:D5=17
130 PRINT#2,A2,B1$;CHR$(29);C1;D1
140 PRINT#2,A2,B2$,CHR$(29),C2;D2
150 PRINT#2,A3,V3$+CHR$(160)+N3$,CHR$(29),C4,D4
160 PRINT#2,A4,S1$+S2$,CHR$(29),C4,D4
170 PRINT#2,A5,O1$+O2$,CHR$(29),C5,D5
180 CMD1:PRINT:PRINT:PRINT:PRINT:LIST
190 REM
200 REM DAS BEISPIEL ZEIGT, WIE VORNULLEN, TEXT, FLIESENDES $ - ZEICHEN UND
210 REM GANZE ZAHLEN BEI UNTERDRUECKTEN VORNULLEN AUSGEDRUCKT WERDEN.
220 REM IN DER 2. DRUCKSPALTE BLEIBT DER LETZTE WERT OHNE AUSDRUCK, WEIL D5=0
230 REM UND DIE FORMATIERUNG 999 KEINE NULLEN ERZWINGT
240 REM DER TEXT IN DER 3. DRUCKZEILE IST UM 1 ZEICHEN ZU LANG, DIE FORMA-
250 REM TIERUNGSANWEISUNG SCHNEIDET DIESES ZEICHEN EINFACH AB
READY.

```


Beispiel 7



```
4000 OPEN4,4
4010 OPEN6,4,6
4020 FORI=1TO12:A$=A$+" "+CHR$(I):NEXT
4030 FORK=1TO2
4040 FORI=1TO8:PRINT#6,CHR$(I)
4050 PRINT#4,A$
4060 NEXT
4070 FORI=8TO1STEP-1:PRINT#6,CHR$(I)
4080 PRINT#4,A$
4090 NEXT
4100 NEXT
4110 PRINT#6,CHR$(24)
4120 CMD4:LIST
500 REM ZEILE 4010 EROEFFNET FILE 6,WELCHER DIE SCHRITTE ZWISCHEN ZWEI
5010 AUSDRUCKZEILEN KONTROLLIERT (NUR CBM 3022)
5020 REM ZEILE 4020 VERDOPPELT BEI JEDEM SCHLEIFENDURCHLAUF DIE
5030 REM SPERRSCHRIFT-MATRIX
READY.
```


Ausdruck des cbm 3022 printers

Beispiel 8




```
10 OPEN4,4
20 FORI=64T090
30 FORJ=1T012:PRINT#4,CHR$(I)CHR$(1);:NEXT
40 PRINT#4
50 NEXT
60 CMD4:LIST
70 REM DURCH DEN SICH 12 X WIEDERHOLENDEN BEFEHL CHR$(1) IN ZEILE 30 WIRD
80 REM JEDESMAL DIE SPERRSCHRIFTMATRIX VERDOPPELT
90 REM ZEILE 20 LEGT DIE BUCHSTABEN FEST (IN ASCII): 64=@ .... 90=Z
READY.
```

Beispiel 9



```
2000 OPEN4,4
2010 OPEN6,4,6
2020 PRINT#6,CHR$(15)
2030 FORI=32TO96:B$=B$+" ":A$=A$+CHR$(I):NEXT:A$=" "A$:B$=" "B$
2040 FORJ=1TO10
2050 PRINT#4,B$:PRINT#4,A$
2070 NEXT
2080 PRINT#4,B$
2090 PRINT#6,CHR$(24)
2100 CMD4:LIST
2110 REM ZEILE 2030 GIBT BLANKS NACH B$ UND BUCHSTABEN (ASCII) NACH A$
20 REM A$ UND B$ ERHALTEN DANN NOCH DAS REVERSE-ZEICHEN
2130 REM ZEILE 2010 EROFFNET DEN FILE 6, WELCHER DIE ANZAHL SCHRITTE ZWISCHEN
2140 ZWEI ZEILEN STEUERT (NUR CBM 3022!)
READY.
```


Beispiel 11



```
10 OPEN4,4
12 OPEN6,4,6
14 PRINT#6,CHR$(14)
15 CMD4
16 FORJ=1TO6
18 I=RND(1)*70
20 PRINTTAB(I)"
22 PRINTTAB(I)" ^
24 PRINTTAB(I)" | |
26 PRINTTAB(I)" | |
28 PRINTTAB(I)" | |
32 PRINTTAB(I)" ^ ^
34 PRINTTAB(I)" | | | |
36 PRINTTAB(I)" ^ ^
38 PRINTTAB(I)" ^
40 PRINTTAB(I)" ^
42 NEXT
44 CMD4:LIST
READY.
```

Beispiel 12

6 ZEILEN/ZOLL

```
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
```

8 ZEILEN JE ZOLL

```
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ
```

```
10 OPEN4,4
20 OPEN6,4,6
30 PRINT#4,CHR$(1)" 6 ZEILEN/ZOLL
40 PRINT#4
50 PRINT#6,CHR$(24)
60 N=6:GOSUB1000
70 PRINT#4
80 PRINT#4,CHR$(1)" 8 ZEILEN JE ZOLL
90 PRINT#4
100 PRINT#6,CHR$(18)
110 N=8:GOSUB1000
120 PRINT#6,CHR$(24)
130 CMD4
140 LIST
150 END
1000 FORI=1TON
1010 PRINT#4,"0123456789ABCDEFGHIJKLMN0PQRSTUVWXYZ"
1020 NEXT
1030 RETURN
READY.
```

ANHANG

Sekundäradresse 6

Durch Angabe der Sekundäradresse 6 wird der Zeilenabstand der cbm 3022 gesteuert. Ein Zoll ist in 144 Schritte aufgeteilt. Der normale Wert ist ein Zeilenabstand von 24 Schritten; dies ist gleichbedeutend 6 Zeilen/Zoll. Die Schritte zwischen den Zeilen werden mit dem Befehl:

```
PRINT#6,CHR$(N)
```

programmiert.

Beispiel:

```
ZEILENABSTAND 24 SCHRITTE  
ZEILENABSTAND 24 SCHRITTE
```

```
10 OPEN4,4:OPEN6,4,6  
20 FORI=1TO10  
30 PRINT#4,"ZEILENABSTAND 24 SCHRITTE"  
40 PRINT#6,CHR$(24)  
50 NEXT  
60 CMD4:LIST  
70 CLOSE4:CLOSE6  
READY.
```

```
10 OPEN4,4:OPEN6,4,6
20 FORI=1TO10
30 PRINT#4,"ZEILENABSTAND 144 SCHRITTE"
40 PRINT#6,CHR$(24)
50 NEXT
53 CLOSE6:CLOSE4:OPEN6,4,6:OPEN4,4
55 PRINT#6,CHR$(144)
60 CMD4:LIST
70 CLOSE4:CLOSE6
READY.
```

ZEILENABSTAND 144 SCHRITTE

ZEILENABSTAND 144 SCHRITTE